<commit_before>INCLUDE(ExternalProject)

SET(NCCL_SOURCE_DIR ${THIRD_PARTY_PATH}/nccl)

INCLUDE_DIRECTORIES(${NCCL_SOURCE_DIR}/src/extern_nccl/src)


if(WITH_DSO)
  # If we use DSO, we do not build nccl, just download the dependencies
  set(NCCL_BUILD_COMMAND "")
  set(NCCL_INSTALL_COMMAND "")
  set(NCCL_INSTALL_DIR "")
else()
  # otherwise, we build nccl and link it.
  set(NCCL_BUILD_COMMAND "make -j 8")
  set(NCCL_INSTALL_COMMAND  "make install")
  SET(NCCL_INSTALL_DIR ${THIRD_PARTY_PATH}/install/nccl)
endif()

ExternalProject_Add(
        extern_nccl
        ${EXTERNAL_PROJECT_LOG_ARGS}
        GIT_REPOSITORY  "https://github.com/NVIDIA/nccl.git"
        GIT_TAG         "v1.3.4-1"
        PREFIX          "${NCCL_SOURCE_DIR}"
        UPDATE_COMMAND  ""
        CONFIGURE_COMMAND ""
        BUILD_COMMAND     "${NCCL_BUILD_COMMAND}"
        INSTALL_COMMAND   "${NCCL_INSTALL_COMMAND}"
        INSTALL_DIR       "${NCCL_INSTALL_DIR}"
        TEST_COMMAND      ""
)

if (WITH_DSO)
  if (${CMAKE_VERSION} VERSION_LESS "3.3.0")
    set(dummyfile ${CMAKE_CURRENT_BINARY_DIR}/lib_any_dummy.c)
    file(WRITE ${dummyfile} "const char * dummy_any = \"${dummyfile}\";")
    add_library(nccl STATIC ${dummyfile})
  else()
    add_library(nccl INTERFACE)
  endif()
else()
  ADD_LIBRARY(nccl STATIC IMPORTED GLOBAL)
  SET_PROPERTY(TARGET nccl PROPERTY IMPORTED_LOCATION
          ${NCCL_INSTALL_DIR}/lib/libnccl.a)
endif()

add_dependencies(nccl extern_nccl)

LIST(APPEND external_project_dependencies nccl)<commit_msg>Correct the install command, static library name and typo in nccl.cmake. (#5048)<commit_after>include(ExternalProject)

set(NCCL_SOURCE_DIR ${THIRD_PARTY_PATH}/nccl)

include_directories(${NCCL_SOURCE_DIR}/src/extern_nccl/src)

if(WITH_DSO)
  # If we use DSO, we do not build nccl, just download the dependencies
  set(NCCL_BUILD_COMMAND "")
  set(NCCL_INSTALL_COMMAND "")
  set(NCCL_INSTALL_DIR "")
else()
  # otherwise, we build nccl and link it.
  set(NCCL_INSTALL_DIR ${THIRD_PARTY_PATH}/install/nccl)
  # Note: cuda 8.0 is needed to make nccl
  # When cuda is not installed on the system directory, need to set CUDA_HOME to your cuda root
  set(NCCL_BUILD_COMMAND "make -j 8")
  set(NCCL_INSTALL_COMMAND  "make install PREFIX=${NCCL_INSTALL_DIR}")
endif()

ExternalProject_Add(
    extern_nccl
    ${EXTERNAL_PROJECT_LOG_ARGS}
    GIT_REPOSITORY  "https://github.com/NVIDIA/nccl.git"
    GIT_TAG         "v1.3.4-1"
    PREFIX          "${NCCL_SOURCE_DIR}"
    UPDATE_COMMAND  ""
    CONFIGURE_COMMAND ""
    BUILD_COMMAND     "${NCCL_BUILD_COMMAND}"
    INSTALL_COMMAND   "${NCCL_INSTALL_COMMAND}"
    INSTALL_DIR       "${NCCL_INSTALL_DIR}"
    TEST_COMMAND      ""
)

if(WITH_DSO)
  if(${CMAKE_VERSION} VERSION_LESS "3.3.0")
    set(dummyfile ${CMAKE_CURRENT_BINARY_DIR}/lib_nccl_dummy.c)
    file(WRITE ${dummyfile} "const char * dummy_nccl = \"${dummyfile}\";")
    add_library(nccl STATIC ${dummyfile})
  else()
    add_library(nccl INTERFACE)
  endif()
else()
  add_library(nccl STATIC IMPORTED GLOBAL)
  set_property(TARGET nccl PROPERTY IMPORTED_LOCATION
               ${NCCL_INSTALL_DIR}/lib/libnccl_static.a)
endif()

add_dependencies(nccl extern_nccl)
{'fraction_non_alphanumeric': 0.10954174513496548, 'fraction_numerical': 0.006905210295040804, 'mean_word_length': 4.050713153724248, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1173467', 'n_tokens_mistral': 1215, 'n_tokens_neox': 1212, 'n_words': 219}